KIT
de sobrevivencia em
CALCULO
Prof. Doherty Andrade - DMA- UEM
doherty@gauss.dma.uem.br

Derivadas Parciais, máximos e mínimos

Aqui temos tres funções muito básicas de duas variáveis:

> **restart:**

> **f:=x^2+0.5*y^2;**

$$f := x^2 + .5\,y^2$$

> **g:=x^2-0.5*y^2;**

$$g := x^2 - .5\,y^2$$

> **h:=-x^2-0.5*y^2;**

$$h := -x^2 - .5\,y^2$$

> **with(plots,display3d);**

$$[\mathit{display3d}]$$

> **setoptions3d(shading=ZHUE,style=patchnogrid);**

$$\mathrm{setoptions3d}(\mathit{shading} = \mathit{ZHUE}, \mathit{style} = \mathit{patchnogrid})$$

>

Aqui está o gráfico de **f** -- assegure-se de que sabe utilizar as facilidades do maple, como girar e incluir eixos no grafico.

> **plot3d(f,x=-5..5,y=-5..5,view=-5..20);**



> **plot3d(g,x=-5..5,y=-5..5,view=-10..10);**



> **plot3d(h,x=-5..5,y=-5..5,view=-20..5);**



Aqui vamos ver tipos básicos de ***pontos critios*** - **f** tem um mínimo na origem, **g** tem um ponto de sela na origem e **h** tem um máximo na origem.

Maple pode tomar as derivadas parciais -- no comando " **diff** " nós dizemos ao Maple em que variavel deve-se diferenciar:

Considere a seguinte função - determinar e classificar os seus pontos criticos:

> **f:=3*x^2-6*x*y+y^3-9*y;**

$$f := 3\,x^2 - 6\,x\,y + y^3 - 9\,y$$

> **fx:=diff(f,x);**

$$fx := 6\,x - 6\,y$$

> **fy:=diff(f,y);**

$$fy := -6\,x + 3\,y^2 - 9$$

> **solve({fx=0,fy=0},{x,y});**

$$\{y = -1, x = -1\}, \{y = 3, x = 3\}$$

Existem dois pontos criticos-- primeiro, vamos ver o gráfico de **f** para ver de que tipo eles são: usamos a variação **x=-3..4** , **y=-3..4** que contém ambos.

> **plot3d(f,x=-3..4,y=-3..4,shading=Z);**



Note que quando **x=y=-1** existe um ponto de sela, e quando **x=y=3** existe um mínimo local:

Para verificar isto, usamos o **teste da derivada segunda** :

> **fxx:=diff(fx,x);**

$$fxx := 6$$

> **fyy:=diff(fy,y);**

$$fyy := 6\,y$$

> **fxy:=diff(fx,y);**

$$fxy := -6$$

> **diff(fy,x);**

$$-6$$

(Isto verifica as derivadas mistas)

> **subs(x=-1,y=-1,fxx*fyy-fxy^2);**

$$-72$$

Esta é negativa, então **(-1,-1)** é sela--

```
> subs(x=3,y=3,fxx*fyy-fxy^2);
```

$$72$$

Esta é positiva, então **(3,3)** ou é um minimo ou um maximo, mas...

```
> subs(x=3,y=3,fxx);
```

$$6$$

```
>
```

Como este é positivo, então **(3,3)** é um minimo.